Arrays – Data Step Efficiency
Harry Droogendyk, Stratia Consulting Inc., Lynden, ON

ABSTRACT
Arrays are a facility common to many programming languages, useful for programming efficiency. SAS® data step arrays have a number of unique characteristics that make them especially useful in enhancing your coding productivity. This presentation will provide a useful tutorial on the rationale for arrays and their definition and use.

INTRODUCTION
Most of the row-wise functionality required for data collection, reporting and analysis is handled very efficiently by the various procedures available within SAS. Column-wise processing is a different story. Data may be supplied in forms that require the user to process across the columns within a single row to achieve the desired result. Column-wise processing typically requires the user to employ the power and flexibility of the SAS data step.

Even within the SAS data step different methods with varying degrees of coding and execution efficiency can be employed to deal with column-wise processing requirements. Data step arrays are unparalleled in their ability to provide efficient, flexible coding techniques to work with data across columns, e.g. a single row of data containing 12 columns of monthly data.

This paper will cover array definition, initialization, use and the efficiencies afforded by their use.

DEFINING ARRAYS
Array definition in SAS is quite different than most other computer languages. In other languages, arrays are typically a temporary container that holds single or multi-dimensional data that has convenient indexing capabilities which may be used to iterate over the data items within, or to directly reference specific items. SAS’s _temporary_ array structures are most like the definitions found in other languages, but SAS provides so much more array functionality than found in other languages.

The basic array definition statement is as follows:

\[
\text{ARRAY array_name (n) <$> (length) array-elements <(initial-values)>;}
\]

- array_name  any valid SAS variable name
- (n) number of array elements (optional in some cases)
- $ indicates a character array
- length length of each array element (optional in some cases)
- array-elements _temporary_ or SAS variable name(s) or variable lists (optional)
- initial-values array initial values

SAS arrays must contain either numeric or character data, not a mixture of both.

TEMPORARY ARRAY DEFINITION
Temporary arrays provide convenient tabular data stores which persist only while the data step is executing. The ‘variables’ created by the temporary array are automatically dropped from the output dataset. Temporary arrays are often used to act as accumulators, store factors etc., when the items being referenced are not required to be stored with the data. Temporary arrays are automatically retained across data step iterations, i.e. the element values are not set to missing at the top of the data step like most other data step variables.

\[
\text{array factors(12,3) _temporary_ ;}
\]
\[
\text{array desc(12) 15 _temporary_ ;}
\]

The first statement defines a two-dimension numeric array. The second, a 12 element character array, each element 15 bytes in length.
PDV VARIABLE ARRAY DEFINITION
Non-temporary data step arrays demonstrate the real flexibility and versatility of SAS arrays. These types of arrays can refer to a series of variables already present in the program data vector (PDV) or even create variables while permitting those variables to be referenced via array structures. Some examples will illustrate the possibilities:

Consider an existing dataset containing four columns of quarterly metrics:

<table>
<thead>
<tr>
<th>cust_id</th>
<th>qtr1</th>
<th>qtr2</th>
<th>qtr3</th>
<th>qtr4</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>175</td>
<td>397</td>
<td>400</td>
<td>261</td>
</tr>
<tr>
<td>2</td>
<td>922</td>
<td>790</td>
<td>543</td>
<td>532</td>
</tr>
<tr>
<td>3</td>
<td>50</td>
<td>67</td>
<td>820</td>
<td>524</td>
</tr>
<tr>
<td>4</td>
<td>854</td>
<td>68</td>
<td>956</td>
<td>298</td>
</tr>
<tr>
<td>5</td>
<td>273</td>
<td>690</td>
<td>977</td>
<td>227</td>
</tr>
<tr>
<td>6</td>
<td>698</td>
<td>413</td>
<td>559</td>
<td>229</td>
</tr>
<tr>
<td>7</td>
<td>476</td>
<td>845</td>
<td>636</td>
<td>551</td>
</tr>
</tbody>
</table>

In the following data step, the ARRAY statement will create qtr1, qtr2, qtr3, qtr4 variables in the PDV as per the array definition since they don't already exist.

```
set quarterly_data;
array q qtr1-qtr4;
  do _i = 1 to 4;
    put cust_id= @13 q(_i)=;
  end;
  cust_id=1 qtr1=185
cust_id=1 qtr2=971
cust_id=1 qtr3=400
cust_id=1 qtr4=260
cust_id=2 qtr1=922
cust_id=2 qtr2=970
cust_id=2 qtr3=543
end

In the following data step, the ARRAY statement will create qtr1, qtr2, qtr3, qtr4 variables in the PDV as per the array definition since they don't already exist.
```

```
data quarterly_data;
  length cust_id 8;
  array q 8 qtr1-qtr4;
  do cust_id = 1 to 7;
    do _i = 1 to 4;
      q(_i) = ceil(ranuni(1) * 1000);
    end;
    output;
  end;
  drop _: ;
run;
```
The following statement also defines the variables, but uses the array name to provide the prefix of the PDV variable name and suffixes the name with consecutive numbers. Again, qtr1, qtr2, qtr3, qtr4 will be created at compile time.

```sas
array qtr(4) 8;
```

The “QTR” array name might look familiar – that’s because it’s also a built-in SAS function. The compiler takes note of that fact and warns the user that the QTR() function is unavailable in this data step because the array reference will take precedence.

NOTE: The array qtr has the same name as a SAS-supplied or user-defined function. Parentheses following this name are treated as array references and not function references.

The first example showed the use of the qtr1 – qtr4 variable list to define the array elements. In the same way, any variable list syntax may be used:

```sas
array desc _character_;       * character vars already defined in PDV;
array amts _numeric_;         * numeric vars already defined in PDV;
```

Arrays may be made up of variables of different lengths. Note the variable list below which specifies that all character variables between flag_x and desc (as defined by the PDV order) be part of the LST array though each character variable has a different length and they are not contiguous.

```sas
length flag_x $10
amt 8
name $20
dol 4
desc $60;
array lst flag_x -character- desc;
```

An instance of a numeric array made up of various numeric variables, i.e. without a common prefix like “qtr”:

```sas
array various discount refund pre_tax after_tax payable received;
```

By default data step arrays are indexed starting at 1, i.e. the first element in the array above would be referenced by various(1). The array size, i.e. the number of elements, may be derived using the DIM() function. However, it’s often advantageous to begin indexing an array with a value other than 1, e.g. in the event we wanted to use the age of school-age children (5-18 years old) directly as an index into an array: An example will make this more clear. Note the use of the various functions which describe the array:

- DIM() which provides the number of array elements
- LBOUND() the index value of the lower array boundary
- HBOUND() the index value of the upper array boundary

```sas
data _null_;       
array ages (5:18) _temporary_;  
d = dim(ages);  
l = lbound(ages);  
h = hbound(ages);  
put ’Ages array has ’ d ’ elements, bounded by ’ l ’ and ’ h’;  
do until(done);  
   set sashelp.class end = done;  
   ages(age) + 1;  
end;  
do i = lbound(ages) to hbound(ages);  
   put i 2. @7 ages(i);  
```
The log results follow, showing the array element ranges and the number of children in each age range.

<table>
<thead>
<tr>
<th>Age</th>
<th>Male</th>
<th>Female</th>
</tr>
</thead>
<tbody>
<tr>
<td>10</td>
<td></td>
<td></td>
</tr>
<tr>
<td>11</td>
<td>1</td>
<td>1</td>
</tr>
<tr>
<td>12</td>
<td>3</td>
<td>2</td>
</tr>
<tr>
<td>13</td>
<td>1</td>
<td>2</td>
</tr>
<tr>
<td>14</td>
<td>2</td>
<td>2</td>
</tr>
<tr>
<td>15</td>
<td>2</td>
<td>2</td>
</tr>
</tbody>
</table>

Multi-dimensional arrays are defined and dealt with in a very similar manner. Note the extra parameter in the DIM() function to reference to the 2nd dimension. In similar fashion, both LBOUND() and HBOUND() may use the 2nd function parameter to refer to specific dimension as well. Log results follow.

```sas
data _null_; array ages(5:18,2) _temporary_; array gender(2) $1 _temporary_ ('M', 'F');

d1 = dim(ages); /* number of elements in 1st dimension;*/
d2 = dim(ages,2); /* number of elements in 2nd dimension;*/
l = lbound(ages);
h = hbound(ages);
put 'Ages array has ' d1 ' x ' d2 ' elements, bounded by ' l ' and ' h;

do until(done);
   set sashelp.class end = done;
   if sex = 'M' then s = 1; else s = 2;
   ages(age,s) + 1;
end;

put 'Age' @6 'Male' @12 'Female';
do i = lbound(ages) to hbound(ages);
   put i @7 ages(i,1) 1. @15 ages(i,2);
end;
stop;
run;
```
Arrays – Data Step Efficiency, continued

16    1       .
17    .       .
18    .       .

ARRAY VALUE INITIALIZATION
Arrays may be initialized at compile time with specific values. Since arrays based on PDV variables are really just a method of referencing “normal” variables, the elements in PDV arrays will be set to missing at the top of the data step as per the usual data step rules.

It’s helpful to assign initial values to the array at compile time since that saves a step during execution. The first two array initialization statements below accomplish the same thing, i.e. (12*0) is the equivalent of twelve zeroes:

data _null_;  
array mths1 (12) _temporary_ (12*0); 
array mths2 (12) _temporary_ (0 0 0 0 0 0 0 0 0 0 0 0); 
array mth_qtr (12) _temporary_ (3*1 3*2 3*3 3*4); 
do i = 1 to 12; 
    put mths1(i)= @ mths2(i)= @ mth_qtr(i)=; 
end; 
run;

<snip>

EFFICIENCIES
Using arrays in the data step will result in coding efficiency, more maintainable code and at times, data-driven code. Data steps that make good candidates for array processing are those that contain "wallpaper code", i.e. a repeating pattern of code where multiple values of the same type are processed, e.g. month1 to month12 values, or calculations involving factors that can be indexed by another data value (e.g. age).

In addition, the SAS summary functions such as SUM() are able to receive an entire array as an input parameter, thus simplifying the code.

STRIPPING WALLPAPER
The most common use for arrays is the ability to iterate over a series of similar variables, performing the same operation on each, without having to code each variable manually.

data monthly_sales; 
set sales; 
if month1 = . then month1 = 0; else month1=round(month1,.1); 
if month2 = . then month2 = 0; else month2=round(month2,.1); 
... 
if month11 = . then month11 = 0; else month11=round(month11,.1); 
if month12 = . then month12 = 0; else month12=round(month12,.1); 
run;

The use of an array reduces the lines of code from 12 to 5:

data monthly_sales; 
set sales; 
array m month1-month12; 
do _i = 1 to dim(m); 
    if m(_i) = . then m(_i) = 0; else m(_i) = round(m(_i),.1); 
end; 
drop _; 
run;
It may be necessary to assign a different factor to a metric depending on a third data item, e.g., dosage variance by age. Rather than coding a series of IF / THEN / ELSE or SELECT / WHEN statements, an array can be employed to lookup the dosage values.

```sas
data dosages;
  set sashelp.class;
  array dosages (5:18) _temporary_ ( .5 .5 .6 .7 1 1.2 1.3 1.5 1.6 1.8 2.1 2.3 2.7 3.0 );
  dosage = dosages(age);
run;

proc print data = dosages noobs;
  var name age dosage;
run;
```

<table>
<thead>
<tr>
<th>Name</th>
<th>Age</th>
<th>dosage</th>
</tr>
</thead>
<tbody>
<tr>
<td>Alfred</td>
<td>14</td>
<td>1.8</td>
</tr>
<tr>
<td>Alice</td>
<td>13</td>
<td>1.6</td>
</tr>
<tr>
<td>Barbara</td>
<td>13</td>
<td>1.6</td>
</tr>
<tr>
<td>Carol</td>
<td>14</td>
<td>1.8</td>
</tr>
<tr>
<td>Henry</td>
<td>14</td>
<td>1.8</td>
</tr>
<tr>
<td>James</td>
<td>12</td>
<td>1.5</td>
</tr>
<tr>
<td>Jane</td>
<td>12</td>
<td>1.5</td>
</tr>
<tr>
<td>Janet</td>
<td>15</td>
<td>2.1</td>
</tr>
<tr>
<td>Jeffrey</td>
<td>13</td>
<td>1.6</td>
</tr>
<tr>
<td>John</td>
<td>12</td>
<td>1.5</td>
</tr>
<tr>
<td>Joyce</td>
<td>11</td>
<td>1.3</td>
</tr>
<tr>
<td>Judy</td>
<td>14</td>
<td>1.8</td>
</tr>
<tr>
<td>Louise</td>
<td>12</td>
<td>1.5</td>
</tr>
<tr>
<td>Mary</td>
<td>15</td>
<td>2.1</td>
</tr>
<tr>
<td>Philip</td>
<td>16</td>
<td>2.3</td>
</tr>
<tr>
<td>Robert</td>
<td>12</td>
<td>1.5</td>
</tr>
<tr>
<td>Ronald</td>
<td>15</td>
<td>2.1</td>
</tr>
<tr>
<td>Thomas</td>
<td>11</td>
<td>1.3</td>
</tr>
<tr>
<td>William</td>
<td>15</td>
<td>2.1</td>
</tr>
</tbody>
</table>

**FUNCTIONS WITH ARRAYS**

Many SAS functions can process an entire array with a very simple coding reference, somewhat akin to how variable lists can be processed by these same functions. In the example below in the second data step, the SUM() function is coded two different ways and there's no real advantage to using an array.

```sas
data sales;
  do cust_id = 1 to 10;
    array mth(12) 0;  * defines mth1 - mth12 to PDV ;
    do _i = 1 to 12;
      mth(_i) = ceil(ranuni(1) * 1000);
    end;
    output;
  end;
  drop _: ;
run;

data monthly_sales;
  set sales;
  array m mth1-mth12;
  annual_sum_array = sum(of m(*) );
run;
```
Arrays – Data Step Efficiency

annual_sum_list = sum(of mth:);
drop mth:;
run;

Results of the array and variable list summation are identical:

<table>
<thead>
<tr>
<th>cust_id</th>
<th>annual_sum_array</th>
<th>annual_sum_list</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>6244</td>
<td>6244</td>
</tr>
<tr>
<td>2</td>
<td>6254</td>
<td>6234</td>
</tr>
<tr>
<td>3</td>
<td>7494</td>
<td>7494</td>
</tr>
<tr>
<td>4</td>
<td>5643</td>
<td>5643</td>
</tr>
<tr>
<td>5</td>
<td>4777</td>
<td>4777</td>
</tr>
<tr>
<td>6</td>
<td>5811</td>
<td>5811</td>
</tr>
<tr>
<td>7</td>
<td>4114</td>
<td>4114</td>
</tr>
<tr>
<td>8</td>
<td>6106</td>
<td>6106</td>
</tr>
<tr>
<td>9</td>
<td>5790</td>
<td>5790</td>
</tr>
<tr>
<td>10</td>
<td>6402</td>
<td>6402</td>
</tr>
</tbody>
</table>

SAS also provides a call routine to sort array contents in ascending order, SORTC and SORTN for character and numeric arrays respectively.

Fred George James Rob Robert Susan george nancy

SPECIAL ARRAY FUNCTIONS

It’s all well and good to define an array that contains a number of character or numeric variables in the PDV. But, sometimes when the array is processed, it’s very helpful to be able to identify the characteristics of the specific variables underlying each array element. There’s a number of functions available, some listed below, that surface the particulars of the variable underlying the array reference. See SAS Online Documentation for a complete list:

<table>
<thead>
<tr>
<th>Function</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>VNAME()</td>
<td>variable name</td>
</tr>
<tr>
<td>VLABEL()</td>
<td>label of the variable</td>
</tr>
<tr>
<td>VFORMAT()</td>
<td>format applied to variable</td>
</tr>
<tr>
<td>VLENGTH()</td>
<td>defined variable length</td>
</tr>
</tbody>
</table>

The real-world example below uses arrays and the VFORMAT function to create a fixed length text file.
Define the skeleton dataset containing the variables and their formats:

```sas
data dialer_skeleton;
  format num $12. name $20.
  expiry $10. msg $15.
  rep $3. ;
  array txtfile _character_; 
  call symputx('dim',dim(txtfile)); * capture array length;
  stop;
run;
```

Define the test data:

```sas
data dialer_input;
  length num $10 name $20 msg $15 rep $2;
  num = '9052941234'; name = 'George Smith'; expiry = '31Oct2012'd;
  msg = '1st call'; rep = '01'; output;
  num = '5196471212'; name = 'Susan Jones'; expiry = '01Nov2012'd;
  msg = 'Second call'; rep = '01'; output;
  num = '4162235678'; name = 'Sam Brown'; expiry = '02Nov2012'd;
  msg = 'New customer'; rep = '2'; output;
  num = '6137219988'; name = 'Brian Tait'; expiry = '29Oct2012'd;
  msg = 'Attriter'; rep = '2'; output;
run;
```

Create output text data. Note the use of the VFORMAT function to retrieve the formatted length to ensure each data item will be written with the correct length to line up properly in the fixed-width text output.

```sas
%macro txt;
  data dialer ( drop = _: );
    if 0 then set dialer_skeleton; * set variable lengths/formats;
    array txtfile _character_; * make available in array;
    set dialer_input ( rename = ( expiry = _exp ));
    expiry = put(_exp,yymmddd10.);
    length record $100;
    record = putc(txtfile(1),vformat(txtfile(1)))
        %do i = 2 %to &dim; * loop limit from skeleton step;
                     || putc(txtfile(&i),vformat(txtfile(&i)))
        %end;
    put record;
  run;
%mend txt;

%txt
```

Results:

```sas
  9052941234  George Smith  2012-10-31  1st call  01
  5196471212  Susan Jones   2012-11-01  Second call 01
  4162235678  Sam Brown     2012-11-02  New customer 2
  6137219988  Brian Tait    2012-10-29  Attriter   2
```
The flexibility of this method comes into play when the text output requirements change. The only code that need be modified is the dialer_skeleton data step. Since the data step that actually creates the text output has no hard-coded variables or lengths, any changes will be picked up automatically. Less maintenance = less errors.

CONCLUSION
SAS data step arrays are simple to define and provide many opportunities for coding efficiency. Common physical data characteristics may be overlaid with logical array definitions which permit the use of looping constructs and functions to write efficient, flexible and maintainable code. Arrays have other uses as well which have been covered by other fine SAS user group papers, e.g. data step transpose. See www.lexjansen.com and search for “array” for additional reading.

REFERENCES
Steve First and Teresa Schudrowitz, “Arrays Made Easy: An Introduction to Arrays and Array Processing”
http://www2.sas.com/proceedings/sugi30/242-30.pdf

http://support.sas.com/documentation/cdl/en/lrcon/68089/HTML/default/viewer.htm#p1i7zivukj8arhn1ihivqz9mot9m.htm

CONTACT INFORMATION
Your comments and questions are valued and encouraged. Contact the author at:
Harry Droogendyk
conf@stratia.ca
Stratia Consulting Inc.
www.stratia.ca

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.
Other brand and product names are trademarks of their respective companies.